Rules for integrands of the form $(a + b\,\text{Sec}[e + f x])^m \left(A + B\,\text{Sec}[e + f x] + C\,\text{Sec}[e + f x]^2\right)$

1: $\int (a + b\,\text{Sec}[e + f x])^m \left(A + B\,\text{Sec}[e + f x] + C\,\text{Sec}[e + f x]^2\right) dx$ when $A b^2 - a b B + a^2 C == 0$

Derivation: Algebraic simplification

Basis: If $A b^2 - a b B + a^2 C == 0$, then $A + B z + C z^2 == \frac{1}{b^2} (a + b z) (b B - a C + b C z)$

Rule: If $a^2 - b^2 \neq 0 \land A b^2 - a b B + a^2 C == 0$, then

$$\int (a + b\,\text{Sec}[e + f x])^m \left(A + B\,\text{Sec}[e + f x] + C\,\text{Sec}[e + f x]^2\right) dx \longrightarrow \frac{1}{b^2} \int (a + b\,\text{Sec}[e + f x])^{m+1} (b B - a C + b C\,\text{Sec}[e + f x])\, dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  1/b^2*Int[(a+b*Csc[e+f*x])^(m+1)*Simp[b*B-a*C+b*C*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C,m},x] && EqQ[A*b^2-a*b*B+a^2*C,0]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  C/b^2*Int[(a+b*Csc[e+f*x])^(m+1)*Simp[-a+b*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,C,m},x] && EqQ[A*b^2+a^2*C,0]
```

2. $\int (b\,\text{Sec}[e + f x])^m \left(A + B\,\text{Sec}[e + f x] + C\,\text{Sec}[e + f x]^2\right) dx$

1. $\int (b\,\text{Sec}[e + f x])^m \left(A + C\,\text{Sec}[e + f x]^2\right) dx$

1: $\int (b\,\text{Sec}[e + f x])^m \left(A + C\,\text{Sec}[e + f x]^2\right) dx$ when $C m + A (m + 1) == 0$

Derivation: Cosecant recurrence 1b with $a \to 0,\ B \to 0,\ C \to -\frac{A (n+1)}{n},\ m \to 0$

Derivation: Cosecant recurrence 3a with $a \to 0,\ B \to 0,\ C \to -\frac{A (n+1)}{n},\ m \to 0$

Rule: If $C m + A (m + 1) == 0$, then

$$\int (b\,\text{Sec}[e+f\,x])^m \left(A+C\,\text{Sec}[e+f\,x]^2\right) dx \longrightarrow -\frac{A\,\text{Tan}[e+f\,x]\,(b\,\text{Sec}[e+f\,x])^m}{f\,m}$$

–

Program code:

```
Int[(b_.*csc[e_.+f_.*x_])^m_.*(A_+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  A*Cot[e+f*x]*(b*Csc[e+f*x])^m/(f*m) /;
FreeQ[{b,e,f,A,C,m},x] && EqQ[C*m+A*(m+1),0]
```

2. $\int (b\,\text{Sec}[e+f\,x])^m \left(A+C\,\text{Sec}[e+f\,x]^2\right) dx$ when $C\,m+A\,(m+1)\neq 0$

   1. $\int (b\,\text{Sec}[e+f\,x])^m \left(A+C\,\text{Sec}[e+f\,x]^2\right) dx$ when $C\,m+A\,(m+1)\neq 0 \wedge m\leq -1$

      1: $\int \text{Sec}[e+f\,x]^m \left(A+C\,\text{Sec}[e+f\,x]^2\right) dx$ when $C\,m+A\,(m+1)\neq 0 \wedge \frac{m+1}{2}\in\mathbb{Z}^-$

Derivation: Algebraic simplification

Basis: If $m\in\mathbb{Z}$, then $\text{Sec}[z]^m \left(A+C\,\text{Sec}[z]^2\right) == \frac{C+A\,\text{Cos}[z]^2}{\text{Cos}[z]^{m+2}}$

Rule: If $C\,m+A\,(m+1)\neq 0 \wedge \frac{m+1}{2}\in\mathbb{Z}^-$, then

$$\int \text{Sec}[e+f\,x]^m \left(A+C\,\text{Sec}[e+f\,x]^2\right) dx \longrightarrow \int \frac{C+A\,\text{Cos}[e+f\,x]^2}{\text{Cos}[e+f\,x]^{m+2}}\,dx$$

–

Program code:

```
Int[csc[e_.+f_.*x_]^m_.*(A_+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  Int[(C+A*Sin[e+f*x]^2)/Sin[e+f*x]^(m+2),x] /;
FreeQ[{e,f,A,C},x] && NeQ[C*m+A*(m+1),0] && ILtQ[(m+1)/2,0]
```

2: $\int (b \operatorname{Sec}[e+f x])^m \left(A+C \operatorname{Sec}[e+f x]^2\right) dx$ when $C m+A (m+1) \neq 0 \land m \leq -1$

Derivation: ???

Rule: If $C m+A (m+1) \neq 0 \land m \leq -1$, then

$$\int (b \operatorname{Sec}[e+f x])^m \left(A+C \operatorname{Sec}[e+f x]^2\right) dx \longrightarrow -\frac{A \operatorname{Tan}[e+f x] (b \operatorname{Sec}[e+f x])^m}{f m}+\frac{C m+A (m+1)}{b^2 m} \int (b \operatorname{Sec}[e+f x])^{m+2} dx$$

Program code:

```
Int[(b_.*csc[e_.+f_.*x_])^m_.*(A_+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  A*Cot[e+f*x]*(b*Csc[e+f*x])^m/(f*m) +
  (C*m+A*(m+1))/(b^2*m)*Int[(b*Csc[e+f*x])^(m+2),x] /;
FreeQ[{b,e,f,A,C},x] && NeQ[C*m+A*(m+1),0] && LeQ[m,-1]
```

2: $\int (b \operatorname{Sec}[e+f x])^m (A+C \operatorname{Sec}[e+f x]^2)\, dx$ when $C m+A (m+1) \neq 0 \land m \nleq -1$

Derivation: Cosecant recurrence 1b with $a \to 0,\ B \to 0,\ m \to 0$

Derivation: Cosecant recurrence 3a with $a \to 0,\ B \to 0,\ m \to 0$

Rule: If $C m+A (m+1) \neq 0 \land m \nleq -1$, then

$$\int (b \operatorname{Sec}[e+f x])^m (A+C \operatorname{Sec}[e+f x]^2)\, dx \longrightarrow \frac{C \operatorname{Tan}[e+f x] (b \operatorname{Sec}[e+f x])^m}{f (m+1)}+\frac{C m+A (m+1)}{m+1} \int (b \operatorname{Sec}[e+f x])^m\, dx$$

Program code:

```
Int[(b_.*csc[e_.+f_.*x_])^m_.*(A_+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cot[e+f*x]*(b*Csc[e+f*x])^m/(f*(m+1)) +
  (C*m+A*(m+1))/(m+1)*Int[(b*Csc[e+f*x])^m,x] /;
FreeQ[{b,e,f,A,C,m},x] && NeQ[C*m+A*(m+1),0] && Not[LeQ[m,-1]]
```

2: $\int (b \operatorname{Sec}[e+f x])^m (A+B \operatorname{Sec}[e+f x]+C \operatorname{Sec}[e+f x]^2)\, dx$

Derivation: Algebraic expansion

Rule:

$$\int (b \operatorname{Sec}[e+f x])^m (A+B \operatorname{Sec}[e+f x]+C \operatorname{Sec}[e+f x]^2)\, dx \longrightarrow \frac{B}{b} \int (b \operatorname{Sec}[e+f x])^{m+1}\, dx+\int (b \operatorname{Sec}[e+f x])^m (A+C \operatorname{Sec}[e+f x]^2)\, dx$$

Program code:

```
Int[(b_.*csc[e_.+f_.*x_])^m_.*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  B/b*Int[(b*Csc[e+f*x])^(m+1),x] + Int[(b*Csc[e+f*x])^m*(A+C*Csc[e+f*x]^2),x] /;
FreeQ[{b,e,f,A,B,C,m},x]
```

3: $\int (a + b\,\mathrm{Sec}[e + f x]) \left(A + B\,\mathrm{Sec}[e + f x] + C\,\mathrm{Sec}[e + f x]^2\right) dx$

Derivation: Algebraic expansion, nondegenerate secant recurrence 1b with $c \to 0,\ d \to 1,\ A \to a\,c,\ B \to b\,c + a\,d,\ C \to b\,d,\ m \to m + 1,\ n \to 0,\ p \to 0$ and algebraic simplification

Basis: $A + B\,z + C\,z^2 == \frac{C\,(d\,z)^2}{d^2} + A + B\,z$

Rule:

$$\int (a + b\,\mathrm{Sec}[e + f x]) \left(A + B\,\mathrm{Sec}[e + f x] + C\,\mathrm{Sec}[e + f x]^2\right) dx \longrightarrow$$

$$\frac{C}{d^2} \int (a + b\,\mathrm{Sec}[e + f x]) (d\,\mathrm{Sec}[e + f x])^2\, dx + \int (a + b\,\mathrm{Sec}[e + f x]) (A + B\,\mathrm{Sec}[e + f x])\, dx \longrightarrow$$

$$\frac{b\,C\,\mathrm{Sec}[e + f x]\,\mathrm{Tan}[e + f x]}{2\,f} + \frac{1}{2} \int \left(2\,A\,a + (2\,B\,a + b\,(2\,A + C))\,\mathrm{Sec}[e + f x] + 2\,(a\,C + B\,b)\,\mathrm{Sec}[e + f x]^2\right) dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -b*C*Csc[e+f*x]*Cot[e+f*x]/(2*f) +
  1/2*Int[Simp[2*A*a+(2*B*a+b*(2*A+C))*Csc[e+f*x]+2*(a*C+B*b)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f,A,B,C},x]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -b*C*Csc[e+f*x]*Cot[e+f*x]/(2*f) +
  1/2*Int[Simp[2*A*a+b*(2*A+C)*Csc[e+f*x]+2*a*C*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f,A,C},x]
```

4: $\int \frac{A + B \operatorname{Sec}[e + f x] + C \operatorname{Sec}[e + f x]^2}{a + b \operatorname{Sec}[e + f x]} \, dx$

Derivation: Algebraic expansion

Basis: $\frac{A+B z+C z^2}{a+b z} == \frac{C z}{b} + \frac{A b+(b B-a C) z}{b (a+b z)}$

Rule:

$$\int \frac{A + B \operatorname{Sec}[e + f x] + C \operatorname{Sec}[e + f x]^2}{a + b \operatorname{Sec}[e + f x]} \, dx \longrightarrow \frac{C}{b} \int \operatorname{Sec}[e + f x] \, dx + \frac{1}{b} \int \frac{A b + (b B - a C) \operatorname{Sec}[e + f x]}{a + b \operatorname{Sec}[e + f x]} \, dx$$

Program code:

```
Int[(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2)/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  C/b*Int[Csc[e+f*x],x] + 1/b*Int[(A*b+(b*B-a*C)*Csc[e+f*x])/(a+b*Csc[e+f*x]),x] /;
FreeQ[{a,b,e,f,A,B,C},x]
```

```
Int[(A_.+C_.*csc[e_.+f_.*x_]^2)/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  C/b*Int[Csc[e+f*x],x] + 1/b*Int[(A*b-a*C*Csc[e+f*x])/(a+b*Csc[e+f*x]),x] /;
FreeQ[{a,b,e,f,A,C},x]
```

5. $\int (a + b \operatorname{Sec}[e + f x])^m \left(A + B \operatorname{Sec}[e + f x] + C \operatorname{Sec}[e + f x]^2\right) dx$ when $a^2 - b^2 == 0$

1: $\int (a + b \operatorname{Sec}[e + f x])^m \left(A + B \operatorname{Sec}[e + f x] + C \operatorname{Sec}[e + f x]^2\right) dx$ when $a^2 - b^2 == 0 \land m < -\frac{1}{2}$

Derivation: Algebraic expansion, singly degenerate secant recurrence 2b with $A \to 1$, $B \to 0$, $p \to 0$ and algebraic simplification

Basis: If $a^2 - b^2 == 0$, then $A + B z + C z^2 == \frac{a A - b B + a C}{a} + \frac{(a + b z)(b B - a C + b C z)}{b^2}$

Rule: If $a^2 - b^2 == 0 \land m < -\frac{1}{2}$, then

$$\int (a + b \operatorname{Sec}[e + f x])^m \left(A + B \operatorname{Sec}[e + f x] + C \operatorname{Sec}[e + f x]^2\right) dx \longrightarrow$$

$$\frac{a A - b B + a C}{a} \int (a + b \operatorname{Sec}[e + f x])^m dx + \frac{1}{b^2} \int (a + b \operatorname{Sec}[e + f x])^{m+1} (b B - a C + b C \operatorname{Sec}[e + f x]) dx \longrightarrow$$

$$\frac{(a A - b B + a C) \operatorname{Tan}[e + f x] (a + b \operatorname{Sec}[e + f x])^m}{a f (2 m + 1)} +$$

$$\frac{1}{a b (2 m + 1)} \int (a + b \operatorname{Sec}[e + f x])^{m+1} (A b (2 m + 1) + (b B (m + 1) - a (A (m + 1) - C m)) \operatorname{Sec}[e + f x]) dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -(a*A-b*B+a*C)*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(a*f*(2*m+1)) +
  1/(a*b*(2*m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*Simp[A*b*(2*m+1)+(b*B*(m+1)-a*(A*(m+1)-C*m))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C},x] && EqQ[a^2-b^2,0] && LtQ[m,-1/2]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -a*(A+C)*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(a*f*(2*m+1)) +
  1/(a*b*(2*m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*Simp[A*b*(2*m+1)-a*(A*(m+1)-C*m)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,C},x] && EqQ[a^2-b^2,0] && LtQ[m,-1/2]
```

2: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx$ when $a^2-b^2=0\,\land\,m\not<-\frac{1}{2}$

Derivation: Nondegenerate secant recurrence 1b with $p\to 0$ and $a^2-b^2=0$

Derivation: Algebraic expansion and singly degenerate secant recurrence 2c with $A\to c,\ B\to d,\ n\to n+1,\ p\to 0$

Basis: $A+B\,z+C\,z^2=C\,z^2+A+B\,z$

Rule: If $a^2-b^2=0\,\land\,m\not<-\frac{1}{2}$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx\,\longrightarrow$$

$$C\int (a+b\,\text{Sec}[e+f\,x])^m\,\text{Sec}[e+f\,x]^2\,dx+\int (a+b\,\text{Sec}[e+f\,x])^m\,(A+B\,\text{Sec}[e+f\,x])\,dx\,\longrightarrow$$

$$\frac{C\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m}{f\,(m+1)}+\frac{1}{b\,(m+1)}\int (a+b\,\text{Sec}[e+f\,x])^m\,(A\,b\,(m+1)+(a\,C\,m+b\,B\,(m+1))\,\text{Sec}[e+f\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(f*(m+1)) +
  1/(b*(m+1))*Int[(a+b*Csc[e+f*x])^m*Simp[A*b*(m+1)+(a*C*m+b*B*(m+1))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,C,m},x] && EqQ[a^2-b^2,0] && Not[LtQ[m,-1/2]]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(f*(m+1)) +
  1/(b*(m+1))*Int[(a+b*Csc[e+f*x])^m*Simp[A*b*(m+1)+a*C*m*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,C,m},x] && EqQ[a^2-b^2,0] && Not[LtQ[m,-1/2]]
```

6. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx$ when $a^2-b^2\neq 0$

1. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx$ when $a^2-b^2\neq 0\land 2\,m\in\mathbb{Z}$

1: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx$ when $a^2-b^2\neq 0\land 2\,m\in\mathbb{Z}^+$

Derivation: Nondegenerate secant recurrence 1b with $p\to 0$

Rule: If $a^2-b^2\neq 0\land m>0$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx\to$$
$$\frac{C\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m}{f\,(m+1)}+$$
$$\frac{1}{m+1}\int (a+b\,\text{Sec}[e+f\,x])^{m-1}\,(a\,A\,(m+1)+((A\,b+a\,B)\,(m+1)+b\,C\,m)\,\text{Sec}[e+f\,x]+(b\,B\,(m+1)+a\,C\,m)\,\text{Sec}[e+f\,x]^2)\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(f*(m+1)) +
  1/(m+1)*Int[(a+b*Csc[e+f*x])^(m-1)*
     Simp[a*A*(m+1)+((A*b+a*B)*(m+1)+b*C*m)*Csc[e+f*x]+(b*B*(m+1)+a*C*m)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f,A,B,C},x] && NeQ[a^2-b^2,0] && IGtQ[2*m,0]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_.*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  -C*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(f*(m+1)) +
  1/(m+1)*Int[(a+b*Csc[e+f*x])^(m-1)*Simp[a*A*(m+1)+(A*b*(m+1)+b*C*m)*Csc[e+f*x]+a*C*m*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f,A,C},x] && NeQ[a^2-b^2,0] && IGtQ[2*m,0]
```

2. $\int (a+b\,\text{Sec}[e+f\,x])^m\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx$ when $a^2-b^2\neq 0\land 2\,m\in\mathbb{Z}^-$

1: $\int \frac{A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$ when $a^2-b^2\neq 0$

Derivation: Algebraic expansion

Basis: $A+B\,z+C\,z^2 == A+(B-C)\,z+C\,z\,(1+z)$

Rule: If $a^2-b^2\neq 0$, then

$$\int \frac{A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx \longrightarrow \int \frac{A+(B-C)\,\text{Sec}[e+f\,x]}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx+C\int \frac{\text{Sec}[e+f\,x]\,(1+\text{Sec}[e+f\,x])}{\sqrt{a+b\,\text{Sec}[e+f\,x]}}\,dx$$

Program code:

```
Int[(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2)/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  Int[(A+(B-C)*Csc[e+f*x])/Sqrt[a+b*Csc[e+f*x]],x] + C*Int[Csc[e+f*x]*(1+Csc[e+f*x])/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,e,f,A,B,C},x] && NeQ[a^2-b^2,0]
```

```
Int[(A_.+C_.*csc[e_.+f_.*x_]^2)/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  Int[(A-C*Csc[e+f*x])/Sqrt[a+b*Csc[e+f*x]],x] + C*Int[Csc[e+f*x]*(1+Csc[e+f*x])/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,e,f,A,C},x] && NeQ[a^2-b^2,0]
```

2: $\int (a+b\,\text{Sec}[e+f\,x])^m\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx$ when $a^2-b^2\neq 0\land 2\,m\in\mathbb{Z}\land m<-1$

Derivation: Nondegenerate secant recurrence 1c with $p\to 0$

Rule: If $a^2-b^2\neq 0\land 2\,m\in\mathbb{Z}\land m<-1$, then

$$\int (a+b\,\text{Sec}[e+f\,x])^m\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx \longrightarrow$$
$$-\frac{(A\,b^2-a\,b\,B+a^2\,C)\,\text{Tan}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m+1}}{a\,f\,(m+1)\,(a^2-b^2)}+$$
$$\frac{1}{a\,(m+1)\,(a^2-b^2)}\int (a+b\,\text{Sec}[e+f\,x])^{m+1}\cdot$$

$$\left(A\left(a^2-b^2\right)(m+1)-a(A b-a B+b C)(m+1)\operatorname{Sec}[e+f x]+\left(A b^2-a b B+a^2 C\right)(m+2)\operatorname{Sec}[e+f x]^2\right) dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  (A*b^2-a*b*B+a^2*C)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(a*f*(m+1)*(a^2-b^2)) +
  1/(a*(m+1)*(a^2-b^2))*Int[(a+b*Csc[e+f*x])^(m+1)*
    Simp[A*(a^2-b^2)*(m+1)-a*(A*b-a*B+b*C)*(m+1)*Csc[e+f*x]+(A*b^2-a*b*B+a^2*C)*(m+2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f,A,B,C},x] && NeQ[a^2-b^2,0] && LtQ[m,-1]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  (A*b^2+a^2*C)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(a*f*(m+1)*(a^2-b^2)) +
  1/(a*(m+1)*(a^2-b^2))*Int[(a+b*Csc[e+f*x])^(m+1)*
    Simp[A*(a^2-b^2)*(m+1)-a*b*(A+C)*(m+1)*Csc[e+f*x]+(A*b^2+a^2*C)*(m+2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,e,f,A,C},x] && NeQ[a^2-b^2,0] && IntegerQ[2*m] && LtQ[m,-1]
```

2: $\int (a+b\operatorname{Sec}[e+f x])^m \left(A+B\operatorname{Sec}[e+f x]+C\operatorname{Sec}[e+f x]^2\right) dx$ when $a^2-b^2\neq 0\land 2m\notin\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $A+B z+C z^2==\frac{A b+(b B-a C) z}{b}+\frac{C z (a+b z)}{b}$

Rule: If $a^2-b^2\neq 0\land 2m\notin\mathbb{Z}$, then

$$\int (a+b\operatorname{Sec}[e+f x])^m \left(A+B\operatorname{Sec}[e+f x]+C\operatorname{Sec}[e+f x]^2\right) dx\longrightarrow$$
$$\frac{1}{b}\int (a+b\operatorname{Sec}[e+f x])^m (A b+(b B-a C)\operatorname{Sec}[e+f x]) dx+\frac{C}{b}\int \operatorname{Sec}[e+f x](a+b\operatorname{Sec}[e+f x])^{m+1} dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  1/b*Int[(a+b*Csc[e+f*x])^m*(A*b+(b*B-a*C)*Csc[e+f*x]),x] + C/b*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1),x] /;
FreeQ[{a,b,e,f,A,B,C,m},x] && NeQ[a^2-b^2,0] && Not[IntegerQ[2*m]]
```

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  1/b*Int[(a+b*Csc[e+f*x])^m*(A*b-a*C*Csc[e+f*x]),x] + C/b*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1),x] /;
FreeQ[{a,b,e,f,A,C,m},x] && NeQ[a^2-b^2,0] && Not[IntegerQ[2*m]]
```

Rules for integrands of the form $(a\,(b\,\mathrm{Sec}[e+f\,x])^p)^m\,\left(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2\right)$

1: $\int(b\,\mathrm{Cos}[e+f\,x])^m\,\left(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2\right)\,dx$ when $m\notin\mathbb{Z}$

Derivation: Algebraic normalization

Basis: $A+B\,\mathrm{Sec}[z]+C\,\mathrm{Sec}[z]^2==\frac{b^2\,\left(C+B\,\mathrm{Cos}[z]+A\,\mathrm{Cos}[z]^2\right)}{(b\,\mathrm{Cos}[z])^2}$

Rule: If $m\notin\mathbb{Z}$, then

$$\int(b\,\mathrm{Cos}[e+f\,x])^m\,\left(A+B\,\mathrm{Sec}[e+f\,x]+C\,\mathrm{Sec}[e+f\,x]^2\right)\,dx\;\longrightarrow\;b^2\int(b\,\mathrm{Cos}[e+f\,x])^{m-2}\,\left(C+B\,\mathrm{Cos}[e+f\,x]+A\,\mathrm{Cos}[e+f\,x]^2\right)\,dx$$

Program code:

```
Int[(b_.*cos[e_.+f_.*x_])^m_*(A_.+B_.*sec[e_.+f_.*x_]+C_.*sec[e_.+f_.*x_]^2),x_Symbol] :=
  b^2*Int[(b*Cos[e+f*x])^(m-2)*(C+B*Cos[e+f*x]+A*Cos[e+f*x]^2),x] /;
FreeQ[{b,e,f,A,B,C,m},x] && Not[IntegerQ[m]]
```

```
Int[(b_.*sin[e_.+f_.*x_])^m_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  b^2*Int[(b*Sin[e+f*x])^(m-2)*(C+B*Sin[e+f*x]+A*Sin[e+f*x]^2),x] /;
FreeQ[{b,e,f,A,B,C,m},x] && Not[IntegerQ[m]]
```

```
Int[(b_.*cos[e_.+f_.*x_])^m_*(A_.+C_.*sec[e_.+f_.*x_]^2),x_Symbol] :=
  b^2*Int[(b*Cos[e+f*x])^(m-2)*(C+A*Cos[e+f*x]^2),x] /;
FreeQ[{b,e,f,A,C,m},x] && Not[IntegerQ[m]]
```

```
Int[(b_.*sin[e_.+f_.*x_])^m_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  b^2*Int[(b*Sin[e+f*x])^(m-2)*(C+A*Sin[e+f*x]^2),x] /;
FreeQ[{b,e,f,A,C,m},x] && Not[IntegerQ[m]]
```

2: $\int (a\,(b\,\text{Sec}[e+f\,x])^p)^m\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx$ when $m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a\,(b\,\text{Sec}[e+f\,x])^p)^m}{(b\,\text{Sec}[e+f\,x])^{m\,p}} == 0$

– Rule: If $m\notin\mathbb{Z}$, then

$$\int (a\,(b\,\text{Sec}[e+f\,x])^p)^m\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx \to$$

$$\frac{a^{\text{IntPart}[m]}\,(a\,(b\,\text{Sec}[e+f\,x])^p)^{\text{FracPart}[m]}}{(b\,\text{Sec}[e+f\,x])^{p\,\text{FracPart}[m]}}\int (b\,\text{Sec}[e+f\,x])^{m\,p}\,(A+B\,\text{Sec}[e+f\,x]+C\,\text{Sec}[e+f\,x]^2)\,dx$$

– Program code:

```
Int[(a_.*(b_.*sec[e_.+f_.*x_])^p_)^m_*(A_.+B_.*sec[e_.+f_.*x_]+C_.*sec[e_.+f_.*x_]^2),x_Symbol] :=
  a^IntPart[m]*(a*(b*Sec[e+f*x])^p)^FracPart[m]/(b*Sec[e+f*x])^(p*FracPart[m])*
    Int[(b*Sec[e+f*x])^(m*p)*(A+B*Sec[e+f*x]+C*Sec[e+f*x]^2),x] /;
FreeQ[{a,b,e,f,A,B,C,m,p},x] && Not[IntegerQ[m]]
```

```
Int[(a_.*(b_.*csc[e_.+f_.*x_])^p_)^m_*(A_.+B_.*csc[e_.+f_.*x_]+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  a^IntPart[m]*(a*(b*Csc[e+f*x])^p)^FracPart[m]/(b*Csc[e+f*x])^(p*FracPart[m])*
    Int[(b*Csc[e+f*x])^(m*p)*(A+B*Csc[e+f*x]+C*Csc[e+f*x]^2),x] /;
FreeQ[{a,b,e,f,A,B,C,m,p},x] && Not[IntegerQ[m]]
```

```
Int[(a_.*(b_.*sec[e_.+f_.*x_])^p_)^m_*(A_.+C_.*sec[e_.+f_.*x_]^2),x_Symbol] :=
  a^IntPart[m]*(a*(b*Sec[e+f*x])^p)^FracPart[m]/(b*Sec[e+f*x])^(p*FracPart[m])*
    Int[(b*Sec[e+f*x])^(m*p)*(A+C*Sec[e+f*x]^2),x] /;
FreeQ[{a,b,e,f,A,C,m,p},x] && Not[IntegerQ[m]]
```

```
Int[(a_.*(b_.*csc[e_.+f_.*x_])^p_)^m_*(A_.+C_.*csc[e_.+f_.*x_]^2),x_Symbol] :=
  a^IntPart[m]*(a*(b*Csc[e+f*x])^p)^FracPart[m]/(b*Csc[e+f*x])^(p*FracPart[m])*
    Int[(b*Csc[e+f*x])^(m*p)*(A+C*Csc[e+f*x]^2),x] /;
FreeQ[{a,b,e,f,A,C,m,p},x] && Not[IntegerQ[m]]
```